

1 Recursion

Recursion is a central concept in F# and is used to control flow in loops without the `for` and `while` constructions. Figure 1.1 illustrates the concept of an infinite loop with recursion.

Figure 1.1: An infinitely long string of "forever forever forever...", conceptually calculated by `let rec forever () = "forever " + (forever ())`.

1.1 Recursive Functions

A *recursive function* is a function which calls itself, and the syntax for defining recursive functions is an extension of that for regular functions:

Listing 1.1: Syntax for defining one or more mutually dependent recursive functions.

```
1 let rec <ident> = <expr> {and <ident> = <expr>} [in] <expr>
```

From a compiler point of view, the `rec` is necessary, since the function is used before the compiler has completed its analysis. If two functions are mutually recursive, then they must be defined jointly using the `and` keyword.

An example of a recursive function that counts from 1 to 10 similarly to `??` is given in Listing 1.2. Here the `prt` function calls itself repeatedly, such that the first call is `prt 1 10`, which calls `prt 2 10`, and so on until the last call `prt 11 10`. Each time `prt` is called, new bindings named `a` and `b` are made to new values. This is illustrated in Figure 1.2. The old values are no longer accessible, as indicated by subscripts in the figure. E.g., in `prt3`, the scope has access to `a3` but not `a2` and `a1`. Thus, in this program, the process is similar to a `for` loop, where the counter is `a`, and in each loop its value is reduced.

The structure of the function is typical for recursive functions. They very often follow

Listing 1.2 countRecursive.fsx:
Counting to 10 using recursion.

```

1  let rec prt a b =
2      if a > b then
3          printf "\n"
4      else
5          printf "%d " a
6          prt (a + 1) b
7
8  prt 1 10

```

```

1  $ fsharp --nologo countRecursive.fsx && mono
    countRecursive.exe
2  1 2 3 4 5 6 7 8 9 10

```

the following pattern.

Listing 1.3: Recursive functions consist of a stopping criterium, a stopping expression, and a recursive step.

```

1  let rec f a =
2      if <stopping condition>
3      then <stopping step>
4      else <recursion step>

```

The *match* – *with* are also very common conditional structures. In Listing 1.2, `a > b` is the *stopping condition*, `printfn "\n"` is the *stopping step*, and `printfn "%d "` `a`; `prt (a + 1) b` is the *recursion step*.

1.2 The Call Stack and Tail Recursion

Fibonacci's sequence of numbers is a recursive sequence of numbers with relations to the Golden ratio and structures in biology. The Fibonacci sequence is the sequence of numbers 1, 1, 2, 3, 5, 8, 13, The sequence starts with 1, 1 and the next number is recursively given as the sum of the two previous ones. A direct implementation of this is given in ???. Here we extended the sequence to 0, 1, 1, 2, 3, 5, ... with the starting sequence 0, 1, allowing us to define all $\text{fib}(n) = 0$, $n < 1$. Thus, our function is defined for all integers, and for the irrelevant negative arguments it fails gracefully by returning 0. This is a general piece of advice: **make functions that**

Listing 1.4 fibRecursive.fsx:

The n 'th Fibonacci number using recursion.

```

1  let rec fib n =
2      if n < 1 then
3          0
4      elif n = 1 then
5          1
6      else
7          fib (n - 1) + fib (n - 2)
8
9  for i = 0 to 10 do
10     printfn "fib(%d) = %d" i (fib i)

```

```

1  $ fsharpc --nologo fibRecursive.fsx && mono
    fibRecursive.exe
2  fib(0) = 0
3  fib(1) = 1
4  fib(2) = 1
5  fib(3) = 2
6  fib(4) = 3
7  fib(5) = 5
8  fib(6) = 8
9  fib(7) = 13
10 fib(8) = 21
11 fib(9) = 34
12 fib(10) = 55

```

fail gracefully.

A visualization of the calls and the scopes created by `fibRecursive` is shown in Figure 1.3. The figure illustrates that each recursive step results in two calls to the function, thus creating two new scopes. And it gets worse. Figure 1.4 illustrates the tree of calls for `fib 5`. Thus, a call to the function `fib` generates a tree of calls that is five levels deep and has `fib(5)` number of nodes. In general for the program in Listing 1.4, a call to `fib(n)` produces a tree with $\text{fib}(n) \leq c\alpha^n$ calls to the function for some positive constant c and $\alpha \geq \frac{1+\sqrt{5}}{2} \sim 1.6$. Each call takes time and requires memory, and we have thus created a slow and somewhat memory-intensive function. This is a hugely ineffective implementation of calculating entries into Fibonacci's sequence, since many of the calls are identical. E.g., in Figure 1.4, `fib 1` is called five times. Before we examine a faster algorithm, we first need to discuss how F# executes function calls.

When a function is called, then memory is dynamically allocated internally for the

function on what is known as the *call stack*. Stacks are used for many things in programming, but typically the call stack is considered special, since it is almost always implicitly part of any program execution. Hence, it is often just referred to as *The Stack*. When a function is called, a new *stack frame* is stacked (pushed) on the call stack, including its arguments, local storage such as mutable values, and where execution should return to when the function is finished. When the function finishes, the stack frame is unstacked (popped) and in its stead, the return value of the function is stacked. This return value is then unstacked and used by the caller. After unstacking the return value, the call stack is identical to its state prior to the call. Figure 1.5 shows snapshots of the call stack when calling `fib 5` in Listing 1.4. The call first stacks a frame onto the call stack with everything needed to execute the function body plus a reference to where the return to, when the execution is finished. Then the body of `fib` is executed, which includes calling `fib 4` and `fib 3` in turn. The call to `fib 4` stacks a frame onto the call stack, and its body is executed. Once execution is returned from the call to `fib 4`, the result of the function is on top of the stack. It is unstacked, saved and the call to `fib 3` is treated equally. When the end of `fib 5` is reached, its frame is unstacked, and its result is stacked. In this way, the call stack is returned to its original state except for the result of the function, and execution is returned to the point right after the original call to `fib 5`. Thus, for Listing 1.4 $\mathcal{O}(\alpha^n)$, $\alpha = \frac{1+\sqrt{5}}{2}$ stacking operations are performed for a call to `fib n`. The $\mathcal{O}(f(n))$ is the *Landau symbol* used to denote the order of a function, such that if $g(n) = \mathcal{O}(f(n))$ then there exists two real numbers $M > 0$ and a n_0 such that for all $n \geq n_0$, $|g(n)| \leq M|f(n)|$. As indicated by the tree in Figure 1.4, the call tree is at most n high, which corresponds to a maximum of n additional stack frames as compared to the starting point.

The implementation of Fibonacci's sequence in Listing 1.4 can be improved to run faster and use less memory. One such algorithm is given in Listing 1.5. Calculating the 45th Fibonacci number a MacBook Pro, with a 2.9 GHz Intel Core i5 using Listing 1.4 takes about 11.2s while using Listing 1.5 is about 224 times faster and only takes 0.050s. The reason is that `fib` in Listing 1.5 calculates every number in the sequence once and only once by processing the list recursively while maintaining the previous two values needed to calculate the next in the sequence. I.e., the function `fibPair` transforms the pair `(a,b)` to `(b,a+b)` such that, e.g., the 4th and 5th pair `(3,5)` is transformed into the 5th and the 6th pair `(5,8)` in the sequence. What complicates the algorithm is that besides the transformation, we must keep track of when to stop, which here is done using a counter variable, that is recursively reduced by 1 until our stopping criterium.

Listing 1.5 also uses much less memory than Listing 1.4, since its recursive call is the last expression in the function, and since the return value of two recursive calls to `fibPair` is the same as the return value of the last. In fact, the return value of any number of recursive calls to `fibPair` is the return value of the last. This structure

Listing 1.5 fibRecursiveAlt.fsx:

A fast, recursive implementation of Fibonacci's numbers. Compare with Listing 1.4.

```

1  let fib n =
2      let rec fibPair n pair =
3          if n < 2 then pair
4          else fibPair (n - 1) (snd pair, fst pair + snd pair)
5      if n < 1 then 0
6      elif n = 1 then 1
7      else fibPair n (0, 1) |> snd
8
9  printfn "fib(10) = %d" (fib 10)

```

```

1  $ fsharp --nologo fibRecursiveAlt.fsx && mono
    fibRecursiveAlt.exe
2  fib(10) = 55

```

is called *tail-recursion*. Compilers can easily optimize the call stack usage for tail recursion, since when in this example `fibPair` calls itself, then its frame is no longer needed, and may be replaced by the new `fibPair` with the slight modification, that the return point should be to `fib` and not the end of the previous `fibPair`. Once the recursion reaches the stopping criteria, then instead of popping a long list of calls of `fibPair` frames, then there is only one, and the return value is equal to the return value of the last call and the return point is to `fib`. Thus, many stack frames in tail recursion are replaced by one. Hence, **prefer tail-recursion whenever possible**. ★

1.3 Mutually Recursive Functions

Functions that recursively call each other are called *mutually recursive* functions. F# offers the `let - rec - and` notation for co-defining mutually recursive functions. As an example, consider the function `even : int -> bool`, which returns true if its argument is even and false otherwise, and the opposite function `odd : int -> bool`. A mutually recursive implementation of these functions can be developed from the following relations: `even 0 = true`, `odd 0 = false`, and for $n > 0$, `even n = odd (n-1)`, which implies that for $n > 0$, `odd n = even (n-1)`: Notice that in the lightweight notation the `and` must be on the same indentation level as the original `let`.

Without the `and` keyword, F# will issue a compile error at the definition of `even`. However, it is possible to implement mutual recursion by using functions as an argu-

Listing 1.6 mutuallyRecursive.fsx:

Using mutual recursion to implement even and odd functions.

```

1  let rec even x =
2      if x = 0 then true
3      else odd (x - 1)
4  and odd x =
5      if x = 0 then false
6      else even (x - 1);;
7
8  let w = 5;
9  printfn "%*s %*s %*s" w "i" w "even" w "odd"
10 for i = 1 to w do
11     printfn "%*d %*b %*b" w i w (even i) w (odd i)

```

```

1  $ fsharp -nologo mutuallyRecursive.fsx && mono
    mutuallyRecursive.exe
2      i  even  odd
3      1 false  true
4      2 true  false
5      3 false  true
6      4 true  false
7      5 false  true

```

ment, e.g., This being said, Listing 1.6 is clearly to be preferred over Listing 1.7.

In the example above, we used the `even` and `odd` function problems to demonstrate mutual recursion. There is, of course, a much simpler solution, which does not use recursion at all: This is to be preferred anytime as the solution to the problem.

1.4 Tracing Recursive Programs

Tracing by hand is a very illustrative method for understanding recursive programs. Consider the recursive program in Listing 1.9. The program includes a function for calculating the greatest common divisor of 2 integers, and calls this function with the numbers 10 and 15. Following the notation introduced in ??, we write:

Step	Line	Env.	Bindings and evaluations
0	-	E_0	$()$
1	1	E_0	$\text{gcd} = ((a, b), \text{gcd-body}, ())$
2	9	E_0	$a = 10$
3	10	E_0	$b = 15$

Listing 1.7 mutuallyRecursiveAlt.fsx:

Mutual recursion without the `and` keyword requires a helper function.

```

1  let rec evenHelper (notEven: int -> bool) x =
2      if x = 0 then true
3      else notEven (x - 1)
4
5  let rec odd x =
6      if x = 0 then false
7      else evenHelper odd (x - 1);;
8
9  let even x = evenHelper odd x
10
11 let w = 5;
12 printfn "%s %s %s" w "i" w "Even" w "Odd"
13 for i = 1 to w do
14     printfn "%d %b %b" w i w (even i) w (odd i)

```

```

1  $ fsharp -nologo mutuallyRecursiveAlt.fsx
2  $ mono mutuallyRecursiveAlt.exe
3      i   Even   Odd
4      1 false   true
5      2 true    false
6      3 false   true
7      4 true    false
8      5 false   true

```

In line 11, `gdc` is called before any output is generated, which initiates a new environment E_1 and executes the code in `gdc-body`:

Step	Line	Env.	Bindings and evaluations
4	11	E_0	<code>gcd a b = ?</code>
5	1	E_1	<code>((a = 10, b = 15), gcd-body, ())</code>

In E_1 we have that $a < b$, which fulfills the first condition in line 2. Hence, we call `gdc` with switched arguments and once again initiate a new environment,

Step	Line	Env.	Bindings and evaluations
6	3	E_1	<code>gcd b a = ?</code>
7	1	E_2	<code>((a = 15, b = 10), gcd-body, ())</code>

In E_2 , $a < b$ in line 2 is false, but $b > 0$ in line 4 is true, hence, we first evaluate `a % b`, call `gcd b (a % b)`, and then create a new environment,

Listing 1.8 parity.fsx:

A better way to test for parity without recursion.

```

1  let even x = (x % 2 = 0)
2  let odd x = not (even x)

```

Listing 1.9 gcd.fsx:

The greatest common divisor of 2 integers.

```

1  let rec gcd a b =
2      if a < b then
3          gcd b a
4      elif b > 0 then
5          gcd b (a % b)
6      else
7          a
8
9  let a = 10
10 let b = 15
11 printfn "gcd %d %d = %d" a b (gcd a b)

```

```

1  $ fsharpc --nologo gcd.fsx && mono gcd.exe
2  gcd 10 15 = 5

```

Step	Line	Env.	Bindings and evaluations
8	5	E_2	$a \% b = 5$
9	5	E_2	$\text{gcd } b (a \% b) = ?$
10	1	E_3	$((a = 10, b = 5), \text{gcd-body}, ())$

Again we fall through to line 5, evaluate the remainder operator and initiate a new environment,

Step	Line	Env.	Bindings and evaluations
11	5	E_3	$a \% b = 0$
12	5	E_3	$\text{gcd } b (a \% b) = ?$
13	1	E_4	$((a = 5, b = 0), \text{gcd-body}, ())$

This time both $a < b$ and $b > 0$ are false, so we fall through to line 7 and return the value of a from E_4 , which is 5:

Step	Line	Env.	Bindings and evaluations
14	7	E_4	return = 5

We scratch E_4 , return to E_3 , either physically or mentally replace the ?-mark with 5 and continue the evaluation of line 5. Since this is also a branch of the last statement in `gcd`, we return the previously evaluated value,

Step	Line	Env.	Bindings and evaluations
15	5	E_3	return = 5

Like before, we scratch E_3 , return to E_2 , either physically or mentally replace the ?-mark with 5 and continue the evaluation of line 5. Since this is also a branch of the last statement in `gcd`, we return the just evaluated value,

Step	Line	Env.	Bindings and evaluations
16	5	E_2	return = 5

Again, we scratch E_2 , return to E_1 , either physically or mentally replace the ?-mark with 5 and continue the evaluation of line 5. Since this is also a branch of the last statement in `gcd`, we return the just evaluated value,

Step	Line	Env.	Bindings and evaluations
17	3	E_1	return = 5

Finally, we scratch E_1 , return to E_0 , either physically or mentally replace the ?-mark with 5 and continue the evaluation of line 5:

3 Step	Line	Env.	Bindings and evaluations
18	11	E_0	output = "gcd a b = 5"
19	11	E_0	return = ()

Note that the output of `printfn` is a side-effect while its return-value is unit. In any case, since this is the last line in our program, we are done tracing.




Figure 1.2: Illustration of the recursion used to write the sequence “1 2 3 ... 10” in line 8 in Listing 1.2. Each frame corresponds to a call to `prt`, where new values overshadow old ones. All calls return `unit`.




Figure 1.3: Illustration of the recursion used to write the sequence “1 2 3 ... 10” in line 8 in Listing 1.2. Each frame corresponds to a call to `fib`, where new values overshadow old ones.




Figure 1.4: The function calls involved in calling `fib 5`.




Figure 1.5: A call to `fib 5` in Listing 1.4 starts a sequence of function calls and stack frames on the call stack.